

連載

Mitaka 改造の手引き (3)

～コマンドセット 2 : 星座～

河村聡人 (Mitaka ワーキンググループ)

1. はじめに

当連載は観測された宇宙を再現するフリーソフト Mitaka (©2005-2023 加藤恒彦, 国立天文台 4 次元デジタル宇宙プロジェクト) のカスタマイズ機能[1]の活用とその普及を目的としています。全 6 回のうち、今回はその 3 回目となります。

連載の予定表

1. 保存と読み込み : 日食 (一地点) [2]
2. コマンドセット 1 : 日食 (多地点) [3]
3. コマンドセット 2 : 星座
4. シーケンス 1 : 月の満ち欠け (宇宙)
5. シーケンス 2 : 月の満ち欠けと日食
6. 字幕と地名 : SPACE-CISTE 簡易版

Mitaka ワーキンググループが行ってきたプログラマー養成イベント(ワークショップ)[4]にて培われた「Mitaka カスタマイズ攻略マニュアル」(旧名:改造マニュアル。以下、教科書)を下書きとしています。ワークショップでは Mitaka 同梱の公式マニュアル[5]を補完または要約した教材を多数整備してきました。それらは次の QR コード先にあります。



図 1 MitakaWG の Google ドライブ[6]

当連載で扱う範囲は Mitaka の利用上の注意[7]に従い、Mitaka の Web サイト[8]にて自由な再配布が認められたものです。なお

Mitaka のスクリーンショットの掲載には別途許可を取っています。

今回は Mitaka の機能の「状態の保存」にて保存した状態のファイルを改造することでコマンドセットを組んでいきました。コマンドセットとは日時や観測地点などを指定する一連の命令をまとめたものでした。

今回は下書きを用意せず、一からコマンドセットを組んでいきます。

2. 開発環境の確認

Mitaka プラグインの開発環境は、バージョン 1.74a の Mitaka を用いて、これまでの連載にて構築済みです。まだ構築していない方は第 1 回 (Mitaka の実行環境の説明) 及び第 2 回 (エディタの説明) を参考にしてください。

3. コマンドセットを組む

基本的にはコマンドセットを一から組むことは推奨されていません。前回行ったように「状態の保存」で保存された命令を下書きすることが望ましいです。しかし、Mitaka の画面上での操作では達成できない表示を行う場合は一から組むこととなります。公式マニュアル mitaka_manual_J.pdf や Mitaka のデフォルトプログラムを参考にしながら組んでいきましょう。

今回はその例として選択した星座のみを表示するプログラムを書きます。表示の切り替えそのものは Mitaka のメニューバーの「表示」>「星座」より、「選択した星座のみ (プラネタリウムモード/宇宙空間モード)」を選択することで行えますが、どの星座を選択す

るかは Mitaka のプログラム上からは行えません。よってカスタマイズの必要があります。

以下、3.1 節から 3.6 節では春の星座の表示／非表示を切り替えるカスタマイズを行います。3.7 節の演習ではそれを拡張し、四季の星座をそれぞれ表示するようにします。また 3.8 節と 3.9 節にて、今回のカスタマイズをより発展させる機能を紹介します。

まずは春の星座の表示／非表示を切り替えるカスタマイズから始めます。既存の mcd ファイルをコピーするか新規のテキストファイルを作成し、spring_constellation.mcd と名前を変更し、エディターで開きます。既存の mcd ファイルをコピーした場合は、内容を全て消してください。

3.1 コマンドセットの定義とメニューバーへの登録

最初にコマンドセットの定義の開始を書きます。前回はファイルを読み込むとすぐに実行・再生されるため、DefCommandSet[STATE] としていました。今回は複数のコマンドセットを定義し、メニューバーから切り替えるようにします。よってコマンドセットに固有の名前を付けます。

DefCommandSet[コマンドセット名, 場所]
KeyString(表示名)

コマンドセット名 にはコマンドセットの固有名 (キー) を入れてください。ここでは春の星座の表示用として ShowCnstSpring とします。

場所 にはメニューバーの登録先を入れます。メニューバーの「プリセット」以下に登録するならば PRESET、「コマンド」以下に登録するならば MENU_BAR を指定します。今回は「コマンド」以下に登録するので、MENU_BAR とします。

KeyString は Mitaka の画面上で表示される文字列を指定します。今回の **表示名** は

春の星座 とします。

ひとつのコマンドセットの定義は次のコマンドセットやシーケンスの定義の開始、またはファイルの末尾までです。

ここまでのファイルを「コマンド定義ファイルを開く」で Mitaka に読み込ませると「コマンド」以下に「春の星座」が追加されていることがわかります (参考例: 図 2)。現状ではコマンドセットの内容が未定義なので、「春の星座」をクリックしても何も起こりません。



図 2 「コマンド」以下に登録されたコマンドセット。3.6 節の演習のコマンド定義ファイルを読み込んだ状態。(Mitaka: ©2005-2023 加藤恒彦, 国立天文台 4 次元デジタル宇宙プロジェクト)

3.2 星座を選択する

Mitaka の programs ディレクトリ内にある default_presets.mcd の 48 行目から 51 行目を参考に、星座の選択を行います。

Mitaka では天体や星座などに固有のキーが振られていますが、それらをまとめて扱うキーグループを設定することで星座の選択が行えます。キーグループの設定は次の命令で行います。

SetKeyGroup(グループキー, キー1, キー2)

この命令はキーを任意のグループキーでまとめるより一般的なものです。星座の選択では **グループキー** を CNST_SELECT に設定しなければなりません。

星座のキーは **CNST_略符** の形式で定義されています。今回は春の星座として、しし座 CNST_Leo、おとめ座 CNST_Vir、うしかい座 CNST_Boo、おおいぬ座 CNST_UMa を設定します。

3.4 選択した星座を表示する。

選択した星座のみを表示するためのスイッチをオンにします。

必要なスイッチのキーを探すには、公式マニュアルの p110 にあるスイッチ一覧が便利です。また起動した Mitaka にて求める表示設定にして、その状態を保存して確認する方法も簡単です。

今回オンにするスイッチは、次の6つです。

- 星座の名前 `CONSTELLATION_NAME`
- 星座線 `CONSTELLATION_LINE`
- 選択した星座のみ（プラネタリウムモード） `CONSTELLATION_SELECT_PLNT`
- 選択した星座のみ（宇宙モード） `CONSTELLATION_SELECT_SPACE`
- 4D2U のロゴ `4D2U_LOGO_DISP`
- クレジット `ON_SCREEN_CREDITS_DISP`

このコマンドセットでは特定のスイッチのみを操作したいので `SwitchOn` コマンドを使用します。間違えて `SwitchSet` コマンドを使用してしまった場合は、指定しなかったスイッチが全てオフになってしまいます。

これで春の星座を表示するコマンドセットはひとまず完成です。

3.5 星座を表示しない

星座の表示を消すコマンドセットを定義します。春の星座を表示するコマンドセットに続けて、新たにコマンドセットの定義 `DefCommandSet[]` を書き込んでいきます。先ほど書いたもののコピーを貼り付けて改造するとより楽です。

コマンドセットのキーを `ShowNoCnst`、表示名を **星座を消す** にします。コマンドセットの内容は `SwitchOff` で星座の表示に関するスイッチをオフにするだけです。

ここまでの内容のコマンド定義ファイルは A.1.1 に掲載しています。Mitaka に読み込んで動作確認をしてください。春の星座の表示

が切り替えられれば正常です。

3.6 他のコマンドセットの呼び出し

同様のコマンドセットを複数定義する場合、共通部分を独立したコマンドセットにして、必要に応じて呼び出すことで使い回すことができます。この方がコード全体としては短くなり、エラーの減少と保守性の向上にもつながります。

次の演習の準備として、3.4 節の部分を独立したコマンドセットに変更します。3.4 節で書いた `SwitchOn` の前の行にコマンドセット `SetCnstVisible` の定義の開始を挿入します。ユーザーの操作から直接実行するものではないので、**コマンドセット名**のみを指定し、**場所**の設定はしません。これで表示の操作が独立したコマンドセットになりました。

春の星座を表示するコマンドセットから独立させた表示のコマンドを実行します。コマンドセット内から他のコマンドセットを実行するには次の命令を使います。

ExecCommandSet(コマンドセット名)

結果的に呼び出し元のコマンドセット自身を呼び出した場合(いわゆる「循環参照」です)、コマンドセットの読み込み処理にループが発生し、抜け出せなくなります。たとえば、コマンドセット A がコマンドセット B を呼び出し、コマンドセット B がコマンドセット A を呼び出すといった具合です。この場合、Mitaka は一定時間操作を受け付けない状態になった後、強制終了します。ご注意ください。

本節の内容を反映したコマンド定義ファイルは A.1.2 に掲載しています。

3.7 演習：四季の星座

四季それぞれの星座を表示するコマンドセットを作成します。これまで作業してきた `spring_constellations.mcd` のコピーを作成

し、seasonal_constellations.mcd と名付けます。

春の星座を表示するコマンドセットと同様に、新たに作成するコマンドセットもメニューバーの「コマンド」に登録してください。

各季節の星座は以下とします。

- 夏の星座：さそり座、いて座、こと座、はくちょう座、わし座
 - 秋の星座：ペガサス座、アンドロメダ座、カシオペア座、みなみのうお座
 - 冬の星座：おうし座、ぎょしゃ座、ふたご座、オリオン座、おおいて座、こいて座
- 本節の演習の解答は A.1.3 に掲載しています。

3.8 発展：ファンクションキーへの登録

Mitaka ではキーボードのキーを押した時の動作もカスタマイズできます。今回メニューバーの「コマンド」に登録した5つのコマンドセットをファンクションキーにも登録すると便利です。F1 はデフォルトでヘルプが登録済みなので F2 以降を使用します。たとえば F2 に春の星座を表示するコマンドセットを登録するには次のコマンドセットをコマンド定義ファイル追加します。

```
DefCommandSet [OnKey_F2]
  ExecCommandSet (SetCnstSpring)
```

3.7 発展：アステリズムの表示

星座と同様に選択したアステリズムのみを表示できます。default_commands.mcd の 61 行目から 70 行目や公式マニュアルの p110 のスイッチ一覧を参考にしてください。現在、Mitaka に実装されているアステリズムは春の大三角、夏の大三角、秋の大四辺形、冬の大三角、冬のダイヤモンド、北斗七星の 6 つです。

3.8 節や 3.9 節の発展的な内容を反映したコマンド定義ファイルは紙幅の都合で割愛いたします。

4. おわりに

今回は複数の状態を遷移させるシーケンスを紹介しました。

文 献

- [1] 加藤恒彦 (2019) 「天文ソフト「Mitaka」の最新機能 ～ユーザーによるカスタマイズ機能と教材開発への応用～」 第 3 3 回天文教育研究会収録, pp157-160
- [2] 河村聡人 (2023) 「Mitaka カスタマイズの手引き (1) ～保存と読み込み：日食 (一地点)～」 天文教育, vol.35, no.3, pp64-69
- [3] 河村聡人 (2023) 「Mitaka カスタマイズの手引き (2) ～コマンドセット 1: 日食 (多地点)～」 天文教育, vol.35, no.4, pp28-33
- [4] 波田野聡美 (2020) 「Mitaka による天文教育／普及ワーキンググループ活動報告」, 第 34 回天文教育研究会収録, pp92-93.
- [5] mitaka 同梱マニュアル：
 - mitaka_manual_J.pdf
 - features_v174.pdf (バージョンに従い名前の一部は変更される)
- [6] Mitaka WG の Google ドライブ
<https://drive.google.com/drive/folders/1vPjyW-j1rEZ0AtGC8I5IrreJSL0eVlX-?usp=sharing>
- [7] Mitaka の利用上の注意
<https://4d2u.nao.ac.jp/t/var/download/notice.html>
- [8] Mitaka ダウンロードページ
<https://4d2u.nao.ac.jp/html/program/mitaka/#Download>



河村 聡人

A.1 ソースコード : コマンド定義ファイル

A.1.1) spring_constellations.mcd : 3.1 節から 3.5 節までの内容

```
1 DefCommandSet[ShowCnstSpring, MENU_BAR]
2   KeyString(春の星座)
3   SetKeyGroup(CNST_SELECT, CNST_Leo, CNST_Vir, CNST_Boo, CNST_UMa)
4   SwitchOn(
5     CONSTELLATION_NAME, CONSTELLATION_LINE,
6     CONSTELLATION_SELECT_PLNT, CONSTELLATION_SELECT_SPACE,
7     4D2U_LOGO_DISP, ON_SCREEN_CREDITS_DISP
8   )
9
10 DefCommandSet[ShowNoCnst, MENU_BAR]
11   KeyString(星座を消す)
12   SwitchOff(
13     CONSTELLATION_NAME, CONSTELLATION_LINE,
14     CONSTELLATION_SELECT_PLNT, CONSTELLATION_SELECT_SPACE
15   )
```

A.1.2) spring_constellations.mcd : 3.6 節の内容を反映

```
1 DefCommandSet[ShowCnstSpring, MENU_BAR]
2   KeyString(春の星座)
3   SetKeyGroup(CNST_SELECT, CNST_Leo, CNST_Vir, CNST_Boo, CNST_UMa)
4   ExecCommandSet(SetCnstVisible)
5
6 DefCommandSet[SetCnstVisible]
7   SwitchOn(
8     CONSTELLATION_NAME, CONSTELLATION_LINE,
9     CONSTELLATION_SELECT_PLNT, CONSTELLATION_SELECT_SPACE,
10    4D2U_LOGO_DISP, ON_SCREEN_CREDITS_DISP
11  )
12
13 DefCommandSet[ShowNoCnst, MENU_BAR]
14   KeyString(星座を消す)
15   SwitchOff(
16     CONSTELLATION_NAME, CONSTELLATION_LINE,
17     CONSTELLATION_SELECT_PLNT, CONSTELLATION_SELECT_SPACE
18  )
```

A.1.3) seasonal_constellations.mcd : 3.7 節の演習の解答

```
1 DefCommandSet[ShowCnstSpring, MENU_BAR]
2   KeyString(春の星座)
3   SetKeyGroup(CNST_SELECT, CNST_Leo, CNST_Vir, CNST_Boo, CNST_UMa)
4   ExecCommandSet(SetCnstVisible)
5
6 DefCommandSet[ShowCnstSummer, MENU_BAR]
7   KeyString(夏の星座)
8   SetKeyGroup(CNST_SELECT, CNST_Sco, CNST_Sgr, CNST_Lyr, CNST_Cyg, CNST_Aql)
9   ExecCommandSet(SetCnstVisible)
10
11 DefCommandSet[ShowCnstFall, MENU_BAR]
12   KeyString(秋の星座)
13   SetKeyGroup(CNST_SELECT, CNST_Peg, CNST_And, CNST_Cas, CNST_PsA)
14   ExecCommandSet(SetCnstVisible)
15
16 DefCommandSet[ShowCnstWinter, MENU_BAR]
17   KeyString(冬の星座)
18   SetKeyGroup(CNST_SELECT,
19     CNST_Ori, CNST_CMa, CNST_CMi, CNST_Gem, CNST_Aur, CNST_Tau)
20   ExecCommandSet(SetCnstVisible)
21
22 DefCommandSet[SetCnstVisible]
23   SwitchOn(
24     CONSTELLATION_NAME, CONSTELLATION_LINE,
25     CONSTELLATION_SELECT_PLNT, CONSTELLATION_SELECT_SPACE,
26     4D2U_LOGO_DISP, ON_SCREEN_CREDITS_DISP
27   )
28
29 DefCommandSet[ShowNoCnst, MENU_BAR]
30   KeyString(星座を消す)
31   SwitchOff(
32     CONSTELLATION_NAME, CONSTELLATION_LINE,
33     CONSTELLATION_SELECT_PLNT, CONSTELLATION_SELECT_SPACE
34   )
```